text: 
  // Peguei aqui: http://forum.arduino.cc/index.php?topic=396450
  // Removi setup e loop
  
  
  // Example 5 - Receive with start- and end-markers combined with parsing
  
  const byte numChars = 32;
  char receivedChars[numChars];
  char tempChars[numChars];        // temporary array for use when parsing
  
  // variables to hold the parsed data
  char messageFromPC[numChars] = {0};
  int integerFromPC = 0;
  float floatFromPC = 0.0;
  
  boolean newData = false;
  
  //============
  
  void recvWithStartEndMarkers() {
    static boolean recvInProgress = false;
    static byte ndx = 0;
    char startMarker = '<';
    char endMarker = '>';
    char rc;
  
    while (Serial.available() > 0 && newData == false) {
      rc = Serial.read();
  
      if (recvInProgress == true) {
        if (rc != endMarker) {
          receivedChars[ndx] = rc;
          ndx++;
          if (ndx >= numChars) {
            ndx = numChars - 1;
          }
        }
        else {
          receivedChars[ndx] = '\0'; // terminate the string
          recvInProgress = false;
          ndx = 0;
          newData = true;
        }
      }
  
      else if (rc == startMarker) {
        recvInProgress = true;
      }
    }
  }
  
  //============
  
  void showParsedData() {
    Serial.print("Message ");
    Serial.println(messageFromPC);
    Serial.print("Integer ");
    Serial.println(integerFromPC);
    Serial.print("Float ");
    Serial.println(floatFromPC);
  }
  
  //============
  
  void parseData() {      // split the data into its parts
  
    char * strtokIndx; // this is used by strtok() as an index
  
    strtokIndx = strtok(tempChars, ",");     // get the first part - the string
    strcpy(messageFromPC, strtokIndx); // copy it to messageFromPC
  
    strtokIndx = strtok(NULL, ","); // this continues where the previous call left off
    integerFromPC = atoi(strtokIndx);     // convert this part to an integer
  
    strtokIndx = strtok(NULL, ",");
    floatFromPC = atof(strtokIndx);     // convert this part to a float
  
  #ifdef DEBUGSERIALINPUT
    showParsedData();
  #endif
  
  }
  